SYSTEM DESIGN AND OPERATION WITH "BASIC"

James B. Grabenstein

POTOMAC VALLEY TELEVISION CO., INC.

This paper will present a list of basic programs that can be used with micro computers for day to day operation of cable tv systems. The programs can be used in micro computers, now on the market. The use of "basic" is a good introduction to learning computer concepts and the digital technicality.

It is not the intention of this paper to come up with any new concepts in computer technology. I will attempt to point out how I got started and some of the standards, I think, the CATV industry needs to follow.

Over the past couple of years I have built several microcomputer systems using the IMSAI 8080 as the base for these systems. In building the IMSAI 8080 systems I have found it extremely difficult to program any practical complex programs using machine language. I find machine language programs very difficult to Debug, consequently, I had to discover a language that was conversational and contained error codes. "Basic" seemed to be the answer.

The local community college was offering a course in "Basic", so I enrolled in it. The course offered theory and terminal time and after a few hours we were working simple programs. Then applying what we had learned to our home terminals programming became a much more understandable and easier to debug.

"Basic" is a fairly simple language to understand and requires little computer theory to understand and operate. It is becoming a standard language in most colleges and high schools throughout the country. So it is not difficult to find classes full or part time in most areas of the country. For the home study enthusiast, many books are available from college book stores, radio and TV supply companies, and micro computers stores. Most of these books are geared as study guides for home study. I have found our local instructors to be as interested in getting involved in "Basic" as we were.

One of the by-products of taking the "Basic" course, I soon found a number of people in our area with interest in microcomputers. Each trying to apply micro technology to their hobby or their business. From this group a local computer club was formed. The club now meets regularly with a constant exchange of ideas and information.

One of the first things we discovered was that we each has a different system. Most of them being built up from componets of varying manufacturers. Each system having different recording or punch systems different Central Processing Unit, in addition each machine's "Basic" had a different dialect or some control code that made exchanging material direct impossible. To exchange programs we would print the program in list form, then correct it for your machine. Then re-insert it in your own system by using a header on the program as to the machine it was originated on, who wrote it, and the amount of memory used. This helped to trace any problems you may encounter. Regular debugging sessions were held at the club or at the class.

Over the past year with the introduction of the TRS-80 Level II System, it has made a package system available to more people in our industry. There are some excellent people writing programs in this format now. I feel we should try to adopt some standard so an easy exchange of programs and date can be established in a uniform format and language exceptable to the majority of our industry assoon as possible. A better method should be the use of frequency shift keying using the seriel input and output of the input/output board. Using the same board rate as your

keyboard with ASCII characters. This system is being developed by Lew Strock,at this time, so we can interface my IMSAI to his TRS-80.

The TRS-80 uses "Microsoft Basic" which is used in a large number of packaged systems now on the market. These systems are well in the price range of most CATV operations and would require a very little dialogue change.

As the system I have used Microsoft Basic and Lew Strock's TRS-80 uses a Microsoft Basic. We wanted to test for any language errors.

To test this interchangeability a program that was written in "Mits Basic" into a 8080 C.P.U. was rewritten by Lew Strock, of Antietam Cable Television, for a TRS 80 Level II. The program was taped and sent to Bob Luff, of the NCTA, for a test run. We had no problem with interchangeability. The program is listed on the next page.

The short programs listed below are part of the original program written in "Mits Basic". Each small program can be entered into most any "Basic" format as a subroutine or as a complete program. These programs are examples of application that can be fuctional in day to day operations of CATV systems. One thing to keep in mind when starting programing a micro computer: Keep your format simple and do not try to program over your head. Try to take one step at a time and programing will fall in place.

The listing on page 4 starting with line 600 is a program intended as training material. The intention is to show the relationship between carrier to noise of an individual piece of equipment to the signal to noise of an entire system. The formulas in this program may have value for other applications.

```
200 REM FIELD INTENSITY (UV/M) WHEN FREQ. & DB ARE KNOWN
201 REM  REF. 0DB=1MV
205 INPUT"DIPOLE LEVEL IN DBMV(0DB=1MV)";D
210 INPUT"WHAT FREQ.(MEG. HZ.)";H
215 LET E1=EXP(LOG(10)*D/20)
216 LET E=1000*E1
217 LET F=.021*E*H
220 PRINT"DIPOLE DB LEVEL";D
225 PRINT"DIPOLE UV LEVEL";E
230 PRINT"UV PER METER   ";F:PRINT:GOTO 205
```

```
260 REM THIS PROGRAM WILL LIST VOLTAGE TO DB LEVELS(0DB=1MV)
265 INPUT"MICROVOLTS";E1
270 E=E1/1000
275 E2=1
280 LET D=20*LOG(E)/LOG(10)
285 PRINT D"DB OR A RATIO OF "E2":"E
290 PRINT
295 GOTO 265
```

```
300 REM NOISE FIGURE OF A CASCADE OF IDENTICAL AMPLIFIERS
315 INPUT"N/F OF OF ONE AMP.----------";F1
320 INPUT"NUMBER OF AMPS. IN CASCADE   ";N
325 LET FM=10*LOG(N)/LOG(10)+F1
330 PRINT N" AMP. NOISE FIGURE"FM" DB"
349 PRINT:GOTO 300
```

```
235 REM FIELD INTENSITY WITH (UV & FREQ. ARE KNOWN
236 PRINT
240 INPUT"UV";E
245 INPUT"MEG.HZ.";H
250 LET F=.021*E*H
255 PRINT"UV/M="F:GOTO 235
```

```
**************************************************************
*                                                            *
*              C A T V  P R O G R A M S                      *
*                                                            *
**************************************************************

1 REM  8080  8K BASIC  MITS(MICROSOFT)
2 REM  16K OF MEMORY  (16384D) (4000H)
3 REM  POTOMAC VALLEY TELEVISION CO.
4 REM  BY JIM GRABENSTIEN
5 REM  UP-DATE: FEBRUARY 1979
6 REM  MODIFIED FOR RADIO SHACK TRS-80 LEVEL II BASIC
7 REM  ANTIETAM CABLE TELEVISION
8 REM  BY LEW STROCK
9 REM  FEBRUARY 1979

100 CLS : PRINT
"***************  C A T V  P R O G R A M S  ***************"
105 PRINT
110 PRINT "1 =  FIELD INTENSITY WITH DB & FREQ KNOWN"   : PRINT
115 PRINT "2 =  FIELL INTENSITY WITH UV & FREQ KNOWN"   : PRINT
120 PRINT "3 =  MICROVOLTS TO DBMV CONVERSION (ODB=1MV)": PRINT
125 PRINT "4 =  NOISE FIGURE FOR IDENTICAL AMP CASCADE" : PRINT
130 PRINT "5 =  SIGNAL TO NOISE OF CATV SYSTEM"         : PRINT
135 PRINT : PRINT
140 INPUT "ENTER THE NUMBER FOR THE PROGRAM YOU WISH TO RUN ";I
145 ON I GOTO  200  , 300  , 400  , 500  , 600

200 REM  FIELD INTENSITY (UV/M) WHEN FREQ. & DB ARE KNOWN
205 REM  REFERENCE ODB = 1MV
210 CLS : PRINT
"*****  F I E L D  I N T E N S I T Y  P R O G R A M S  *****"
215 PRINT : PRINT
220 INPUT "ENTER DIPOLE LEVEL IN DBMV  (OLB=1MV) ";D
225 PRINT
230 INPUT "ENTER THE MEASUREMENT FREQ IN MHZ     ";H
235 PRINT : PRINT : PRINT
240 LET E1 = EXP (LOG (10) * D / 20)
245 LET E = 1000 * E1
250 LET F = .021 * E * H
255 PRINT "FOR A DIPOLE LEVEL OF "; D ;"DBMV AT"; H ;"MHZ"
260 PRINT
265 PRINT "THE CALCULATED FIELD INTENSITY IS"; F ;"UV/METER"
270 PRINT : PRINT : GOSUB 900
275 GOTO 200

300 REM  FIELD INTENSITY WHEN UV & FREQ ARE KNOWN
305 CLS : PRINT
"***** F I E L D  I N T E N S I T Y  P R O G R A M  *****"
310 PRINT : PRINT
315 INPUT "ENTER DIPOLE LEVEL IN UV          ";E
320 PRINT
325 INPUT "ENTER THE MEASUREMENT FREQ IN MHZ ";H
330 PRINT : PRINT : PRINT
335 LET F = .021 * E * H
340 PRINT "FOR A DIPOLE LEVEL OF"; E ;"UV AT"; H ;"MHZ"
345 PRINT
350 PRINT "THE CALCULATED FIELD INTENSITY IS"; F ;"UV/METER"
355 PRINT : PRINT : GOSUB 900
360 GOTO 300
```

```
400 REM THIS PROGRAM WILL CONVERT VOLTAGE LEVELS TO DBMV LEVELS
405 CLS : PRINT
"**   V O L T A G E   T O   D B M V   C O N V E R SII O N   **"
410 PRINT : PRINT
415 INPUT "ENTER LEVEL IN MICROVOLTS ";E1
420 PRINT : PRINT : PRINT
425 E = E1 / 1000
430 E2 = 1
435 LET D = 20 * LOG (E) / LOG (10)
440 PRINT E1 ;"MICROVOLTS = "; D ;"DBMV"
445 PRINT
450 PRINT " WHEN 0 DB IS REFERENCED TO 1000 MICROVOLTS"
455 PRINT
460 PRINT ABS (D) ;" DB EXPRESSES A RATIO OF "; E2 ":" E
465 PRINT : PRINT : GOSUB 900
470 GOTO 400

500 REM  NOISE FIGURE OF A CASCADE OF IDENTICAL AMPLIFIERS
505 CLS : PRINT
"**   NOISE FIGURE OF A CASCADE OF IDENTICAL AMPLIFIERS   **"
510 PRINT : PRINT
515 INPUT "ENTER THE NOISE FIGURE OF ONE AMP   ";F1
520 PRINT
525 INPUT "ENTER THE NUMBER OF AMPS IN CASCADE ";N
530 PRINT : PRINT
535 LET FM = 10 * LOG (N) / LOG (10) + F1
540 PRINT "USING AMPS WITH A NOISE FIGURE OF"; F1 ;"DB"
545 PRINT
550 PRINT"A";N;"AMP CASCADE EXHIBITS A NOISE FIGURE OF";FM;"DB"
555 PRINT : PRINT : GOSUB 900
560 GOTO 500

600 REM  SIGNAL/NOISE AND CARRIER/NOISE OF A CATV SYSTEM
605 CLS : PRINT
"*****   CARRIER/NOISE & SIGNAL/NOISE OF A CATV SYSTEM   *****"
610 PRINT : PRINT
"ASSUME THE SIGNAL/NOISE OF THE TV STATION VIDEO TO BE 49 DB"
615 PRINT
620 INPUT "ENTER THE PRE-AMP INPUT LEVEL IN DBMV ";D
625 INPUT "ENTER THE PRE-AMP NOISE FIGURE IN DB  ";N
630 GOSUB 800
635 PRINT : PRINT
ABS (CN) ;"DB C/N OR"; ABS (SN) ;"DB S/N AT THE PRE-AMP OUTPUT"
640 PRINT
645 INPUT "ENTER THE PROCESSOR INPUT LEVEL IN DBMV ";D
650 INPUT "ENTER THE PROCESSOR NOISE FIGURE IN DB  ";N
655 GOSUB 810
660 PRINT : PRINT
ABS (CN);"DB C/N OR"; ABS (SN);"DB S/N AT THE PROCESSOR OUTPUT"
665 PRINT
670 INPUT "ENTER INPUT LEVEL TO THE FIRST AMP IN CASCADE ";I
675 INPUT "ENTER THE NOISE FIGURE FOR THE LINE AMP       ";N1
680 LET CN = (-59 + N1) -I : GOSUB 820
685 LET CQ = CN
690 INPUT "ENTER THE NUMBER OF AMPS IN CASCADE           ";X
695 CLS : PRINT "AMP # ","RF C/N AMP","VIDEO  S/N" : PRINT
700 FOR M = 1 TO X
705 LET F1 = 10 * LOG (M) / LOG (10)
710 LET CN = CQ + F1
```

```
715 GOSUB 830
720 F1$ = "###" : F2$ = "##.####
725 PRINT USING F1$; M ;: PRINT " ",: PRINT USING F2$; ABS (CN);
    :PRINT " DB",: PRINT USING F2$; ABS(SN);: PRINT " DB"
730 NEXT M
735 PRINT : GOSUB 900
740 GOTO 600

800 LET SN = -49
810 LET CN = (-59 + N) - D
820 LET RS = EXP (LOG (10) * SN / 10)
830 LET RC = EXP (LOG (10) * CN / 10)
840 LET SN = 10 * (LOG (RS + RC) / LOG (10))
850 RETURN

900 C$ = " "
910 PRINT "PUSH ENTER TO CONTINUE USING THIS PROGRAM OR"
920 INPUT "ENTER AN X TO RETURN TO THE SELECTION SCREEN  ";C$
930 IF C$ = "X" GOTO 1 ELSE RETURN
940 END
```

****

These programs are an example of what can be done when industry and individuals start to exchange programs and ideas. A few words of caution. Programs should be structured so they can be followed in step by step order. Keep them simple. Check them for accuracy. Give credit to the program's originator.

One reason I feel that I can give a vote to standardizing on the TRS-80 Level II or a similar format, I know the problems I have encountered with a system that has no direct interchangeability.

If you are starting into a system, start simple and have interchangeability with someone and be prepared to spend many hours. Micro computers have unlimited possibilities but it takes time, patience, and a lot of hard work to make them productive. Your knowledge of digital technology will grow with each problem you solve.

Any program is only as good as the data entered and accuracy of the program itself. Another thing you will find is you will never be satisfied with a program and you will want something changed. I find this in my own programs or one I obtain from other sources.

I would like to thank Lew Strock for the time and effort he put into coverting the "Mit" programs into Level 11 Basic and Bob Luff in cross checking the tapes.

The program on page 6 is an example of material from industry. These programs are very handy to have at your finger tips like any other reference material. If you had to design programs of this caliber each time you needed them, your time would be spent programming and not working A good exchange of programs is needed and Ray Daly is working towards this goal.

I am in favor of such exchanges and will back any other efforts to the best of my ability. Many such high quality programs written in basic for larger systems are available. These programs are very easily adapted to micro computer systems. It is my hope we can excourage you, who have materials, to share them with the CATV industry.

References:

"Technical Handbook for CATV" 3 Addition March 68

Ken Simmons

"Programming with Basic"

Byron S. Gottfried

"Introduction to Computing A Basic Approach"

Richard E. Esposito

Another test that was made a program was obtained from Hughs Microwave. Thanks to Abe Sonnenschein and Norman Woods. This program was written for a Wang Computer and was formatted by Lew Strock for the TRS-80 Level II.

```
100 REM  THIS PROGRAM WILL COMPUTE DISTANCE AND AZIMUTH
110 REM  GIVEN LONGITUDE AND LATITUDE OF TWO POINTS
120 REM  FORMATTED FOR  TRS - 80   16K   LEVEL II
130 REM  LEW STROCK - ANTIETAM CABLE TELEVISION
140 REM                HAGERSTOWN, MARYLAND
150 REM  UPDATE : FORMAT AND DISPLAY 02/24/79
160 REM  ADAPTED FROM A PROGRAM IN WANG BASIC FROM USED MICROWAVE
170 REM  DIMENSION STRINGS AND DEFINE DOUBLE PRECISION
180 DIM A$ (100), B$ (100)
190 DEFDBL A-L
200 REM  CLEAR SCREEN AND GET XMTR SITE NAME AND LOCATION
210 CLS:PRINT TAB(15)"D I S T A N C E  /  A Z I M U T H":PRINT
220 INPUT "TRANSMITTER LOCATION          ";A$
230 REM  CONVERT DEGREES-MINUTES-SECONDS TO ONE NUMBER
240 INPUT "TRANSMITTER LATITUDE -  D,M,S ";D,M,S
250 L1 = D + ( M + (S/60)) / 60
260 INPUT "TRANSMITTER LONGITUDE - D,M,S ";D,M,S
270 L2 = D + ( M + (S/60)) / 60
280 REM  GET RCVR SITE NAME AND LOCATION
290 PRINT
300 INPUT "RECEIVER LOCATION          ";B$
310 INPUT "RECEIVER LATITUDE -  D,M,S ";D,M,S
320 L3 = D + ( M + (S/60)) / 60
330 INPUT "RECEIVER LONGITUDE - D,M,S ";D,M,S
340 L4 = D + ( M + (S/60)) / 60
350 REM  FIND ANGLES  A - B - C  AND CONVERT TO RADIANS
360 A = L2 - L4  :  A = A * .0174532925
370 B = 90 - L1  :  B = B * .0174532925
380 C = 90 - L3  :  C = C * .0174532925
390 REM  COMPUTE DISTANCE AND AZIMUTH USING TRIG
400 E = COS(B) * COS(C) + SIN(B) * SIN(C) * COS(A)
410 F = ATN (SQR (ABS (1-E[2)) / E)
420 G = (COS(C) - COS(F) * COS(B)) / (SIN(F) * SIN(B))
430 H = ATN (SQR (ABS (1-G[2)) / G)
440 REM  CONVERT H BACK TO DEREES
450 H = H * 57.2957795785523
460 IF H > = 0 THEN 490
470 H = H + 180
480 REM  CONVERT F BACK TO DEGREES
490 F = F * 57.2957795785523
500 REM  FIND DISTANCE BY KNOWN DISTANCE SCALE OF CO-ORD
510 I = F * 69.05
520 J = SIN(A)
530 REM  CONVERT J BACK TO DEGREES
540 J = J * 57.2957795785523
550 REM  DISPLAY RESULTS OF COMPUTATIONS
560 PRINT : PRINT A$ ;" TO "; B$ : PRINT
570 PRINT "DISTANCE = "; USING "###.##" ; I ;: PRINT "  MILES";
580 IF J > 0 THEN 600
590 H = 360 - H
600 PRINT "         ";
610 PRINT "AZIMUTH = ";USING "###.##"; H ;: PRINT "  DEGREES"
620 PRINT
630 REM  GO BACK TO GET OTHER POINTS SERVED BY XMTR
640 REM  OR START OVER WITH A NEW XMTR SITE
650 Q$ = " "  ' RESET Q$
660 INPUT"ENTER AN X TO CONTINUE USING THE SAME XMTR SITE"; Q$
670 IF Q$ = "X" THEN PRINT @ 320,; : PRINT CHR$(31) : GOTO 300
680 GOTO 200
690 END
```